

## MYOB Company File: Authentication: MYOB Authentication - OAuth 2.0 Complete

---

#Created: Andrew Duncan - 10 June 2021

#Last Modified: Andrew Duncan - 21 August 2024

#Purpose: complete the MYOB OAuth 2.0 authentication flow. Performed after user has initiated OAuth 2.0 and provided permission for App

#Must be performed on FileMaker Pro/Go client as requires user to login to MYOB Account via WebViewer

#<https://developer.myob.com/api/accountright/api-overview/authentication/>

#Mandatory Checks

**Set Variable** [ \$response; Value:GetLayoutObjectAttribute ( "myobAuthWebViewer" ; "source" ) ]

**Set Field** [ Interface::zv\_LastAPIResult ; \$response ]

#Check for Code in Response URL, otherwise exit

#Code a temporary code that may only be exchanged once and expires 2 minutes after issuance

**If** [ not PatternCount ( \$response ; "code=" ) ]

**Set Variable** [ \$errorText; Value:"The Code Parameter was not received from the MYOB API - please try authenticating again" ]

**Beep**

**Show Custom Dialog** [ Title: "OAuth 2.0 Authentication Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Exit Script** [ Result: \$errorText ]

**Else**

**Set Variable** [ \$code; Value:Let ( [  
t1 = \$response ;  
n1 = Position ( t1 ; "code=" ; 1 ; 1 ) ;  
n2 = Position ( t1 ; "&scope=" ; 1 ; 1 ) ;  
  
t2 = Middle ( t1 ; n1 + 5 ; n2 - n1 - 5 )

];

t2

) // Let  
]

**End If**

**If** [ IsEmpty ( \$code ) ]

*#code Parameter/variable not received*

**Set Variable** [ \$errorText; Value:"The Code URL Parameter was not received - please try authenticating again" ]

**Beep**

**Show Custom Dialog** [ Title: "Authentication Code Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Exit Script** [ Result: \$errorText ]

**End If**

#Check State Parameter matches

**If** [ PatternCount ( \$response ; "state=" ) ]

**Set Variable** [ \$state; Value:Let ( [  
t1 = \$response ;  
n1 = Position ( t1 ; "state=" ; 1 ; 1 ) ;  
n2 = Length ( t1 ) ;

t2 = Right ( t1 ; n2 - n1 - 5 )

];

t2

) // Let ]

If [ \$state ≠ Get (PersistentID) ]

**Set Variable** [ \$errorText; Value:"The State Parameter from MYOB does not match the supplied State from FileMaker - please try authenticating again"

// The state parameter should be used to avoid forgery attacks. Pass in a value that's unique to the user you're sending through authorisation. It will be passed back after the user completes authorisation ]

**Beep**

**Show Custom Dialog** [ Title: "OAuth 2.0 Authentication Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Exit Script** [ Result: \$errorText ]

End If

Else

**Set Variable** [ \$errorText; Value:"The State Parameter was not received - please try authenticating again" ]

**Beep**

**Show Custom Dialog** [ Title: "OAuth 2.0 Authentication Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Exit Script** [ Result: \$errorText ]

End If

*#Exchange the code for an access token*

*#NATIVE STEPS - FILEMAKER V16 OR HIGHER*

**Allow User Abort** [ Off ]

*#Reset the debugging global variables*

**Set Variable** [ \$\$cURLTrace; Value:"" ]

*#Set POST URL*

**Set Variable** [ \$url; Value:"https://secure.myob.com/oauth2/v1/authorize" ]

*#Set cURL options*

**Set Variable** [ \$cURL; Value:"--request POST" &  
"--trace-ascii \$\$cURLTrace" &

"--header " & Quote( "Content-type" & ": " & "application/x-www-form-urlencoded" ) &

"--data " & Quote( "code=" & \$code ) &

"--data " & Quote( "redirect\_uri=" & Interface::RedirectURL ) &

"--data " & Quote( "scope=CompanyFile" ) &

"--data " & Quote( "client\_id=" & Interface::APIKey ) &

"--data " & Quote( "client\_secret=" & Interface::APISecret ) &

"--data " & Quote( "grant\_type=authorization\_code" ) &

"--dump-header \$responseHeaders"

&

Case (

not IsEmpty ( Interface::ProxyServerAddress ) ;

"--proxy " & Interface::ProxyServerAddress & ":" & Interface::ProxyServerPort &

"--proxy-user " & Interface::ProxyServerUserName & ":" & Interface::ProxyServerPassword

) ]

*#Perform HTTP REQUEST*

## MYOB Company File: Authentication: MYOB Authentication - OAuth 2.0 Complete

---

**Set Error Capture** [ On ]

**Insert from URL** [ \$response; \$url; cURL options: \$cURL ]  
[ Select; No dialog ]

**Set Variable** [ \$lastError; Value:Get (LastError) ]

**If** [ \$lastError ≠ 0 ]

*#Insert from URL Error*

**Set Variable** [ \$errorText; Value:"There was an error performing the Insert from URL HTTP Request. Error: " & \$lastError & " (" & ErrorDescription ( \$lastError ) & ")" ]

**Beep**

**Show Custom Dialog** [ Title: "HTTP Request Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Set Field** [ Interface::zv\_LastAPIResult\_gt; \$errorText ]

**Exit Script** [ Result: \$errorText ]

**End If**

*#Check Response Code*

**Set Variable** [ \$responseCode; Value:httpResponseCode ( \$responseHeaders ) ]

*#Check for Errors*

**Set Variable** [ \$error; Value:JSONGetElement ( \$response ; "error" ) ]

**If** [ \$responseCode ≠ 200 ]

*#HTTP Response Code <> 200*

**Set Variable** [ \$errorText; Value:"There was an error performing the HTTP Request. HTTP Response Code: " & \$responseCode & " Error: " & \$error & " " & \$errorAdditionalDetails ]

**Beep**

**Show Custom Dialog** [ Title: "HTTP Request Error"; Message: \$errorText; Default Button: "OK", Commit: "Yes" ]

**Set Field** [ Interface::zv\_LastAPIResult\_gt; \$errorText ]

**Exit Script** [ Result: \$errorText ]

**Else**

*#OK - just record the response*

**Set Variable** [ \$response; Value:JSONFormatElements ( \$response ) ]

**Set Field** [ Interface::zv\_LastAPIResult\_gt; \$response ]

**End If**

*#Process HTTP Response*

**Set Variable** [ \$accessToken; Value:JSONGetElement ( \$response ; "access\_token" ) ]

**Set Variable** [ \$expiresIn; Value:JSONGetElement ( \$response ; "expires\_in" ) ]

**Set Variable** [ \$refreshToken; Value:JSONGetElement ( \$response ; "refresh\_token" ) ]

*#Set fields with token details and keys*

**Set Field** [ Interface::MYOBAccessToken; \$accessToken ]

**Set Field** [ Interface::MYOBRefreshToken; \$refreshToken ]

**Set Field** [ Interface::AccessTokenExpires; Get (CurrentHostTimestamp) + \$expiresIn ]

*#Clear out the Web Viewer URL field*

**Set Field** [ Interface::zl\_WebViewerURL\_gt; "" ]

**Commit Records/Requests**

*#Close Card Window*

**Close Window** [ Current Window ]

**Beep**

**Show Custom Dialog** [ Title: "Authentication Successful"; Message: "You have successfully authenticated with MYOB"; Default Button: "OK", Commit: "Yes" ]

**Exit Script** [ ]

*#History:*

*#v1.00: Andrew Duncan - 10 June 2021*

*#v1.01: Andrew Duncan - 21 August 2024*

*#- updated for changes to MYOB API Cloud Authentication*